

# SIMULACIÓ I ANÀLISI DE CIRCUITS MITJANÇANT PSPICE

## Sessions 5 i 6 PSPICE

Simulació a partir de Netlist  
Creació de Símbols  
(OrCad Capture)

Versió 2 - Febrer 2011

Juan A. Chávez, Santiago Silvestre, Antoni Turó

## Introducció

En aquestes sessions s'aprofundirà en l'ús del simulador Pspice AD incidint en les possibilitats que ofereix a nivell d'entrada per esquemàtic i/o netlist i les prestacions del programa OrCAD Capture per definir parts i símbols específics per a les nostres necessitats. Amb aquesta finalitat utilitzarem com exemple el circuit equivalent d'una cèl·lula solar. Finalment veurem com es pot realitzar una anàlisi de Fourier a partir de la simulació d'un senyal modulad AM.

## Objectiu

Els objectius d'aquestes dues sessions és que els estudiants coneguin les següents característiques d'aquest software de simulació :

- 1- Entrada a simulació Pspice AD per Netlist
- 2- Possibilitat de generació de símbols i parts que permeten facilitar l'entrada de circuits predefinit a la finestra d'esquemàtics d'OrCAD Capture.
- 3- Finalment, veure el contingut espectral d'un senyal mitjançant la transformada de Fourier.

## Generació i anàlisi del Netlist

En aquest apartat entrarem un circuit senzill utilitzant la finestra d'entrada d'esquemàtics de l'OrCAD Capture. Amb aquest objectiu, cal crear un nou projecte de la mateixa forma que s'ha fet en les sessions anteriors. El circuit que cal entrar en l'esquemàtic és el de la figura 1.




Figura 1. Proposta d'esquemàtic.

Un cop introduït l'esquemàtic, heu de configurar el perfil de la simulació amb: **Pspice > NewSimulation Profile**, o bé clickant la icona equivalent del menú **Pspice**.

Després de donar nom al perfil, heu de configurar-lo amb una anàlisi del tipus **"Time domain Transient"**, i escollir a Options : **Parametric Sweep**.

A Sweep variable cal escollir “ **Voltage source** “, amb el nom de V1. Fixem ara Sweep type “ **Linear** “, amb un valor “Start Value” de **1 V** i a “End Value” igual a **5 V** i el pas de tensió a “Increment” igual a **0.5 V**.

Executeu la simulació i a la finestra de Pspice, **Available Sections**, escolliu **All** i feu **OK**. Per veure els resultats de la simulació fem **Trace > Add Trace** i escollim la variable **I(R3)**. Això ens genera un plot com el de la figura 2, on podem veure els diferents valors del corrent per cadascun dels valors del paràmetre V1 que hem entrat en la simulació.




Figura 2. Resultats primera simulació.

Si ara tornem a la finestra de OrCAD Capture, podem escollir **Pspice > Create Netlist**, amb aquesta opció el programa genera el fitxer text corresponent a l'esquemàtic que s'anomena Netlist. Per veure el contingut d'aquest fitxer fem ara **Pspice > View Netlist**, El programa “ Bloc de Notes” ens mostrarà el text corresponent a aquest Netlist, que ha de ser semblant al següent (pot canviar l'ordre dels components i el número dels nodes):

```
* source

R_R1    N00169 N00148 1k TC=0,0

V_V1    N00169 0 0Vdc

R_R2    N00148 0 1k TC=0,0

G_G1    N00310 0 N00148 0 10

R_R3    0 N00310 10 TC=0,0
```

A cada fila d'aquest fitxer apareix en primer lloc el tipus de component i el seu nom, seguidament els dos nodes entre els que s'ha connectat, i finalment el seu valor o el valor dels seus paràmetres. Compareu el contingut del Netlist amb l'esquemàtic dibuixat a OrCAD Capture.

Per conèixer millor les opcions que ens ofereix el Netlist ara generarem un nou projecte a OrCAD Capture que ens permeti reproduir el circuit elèctric equivalent d'una cèl·lula solar. Amb aquest objectiu, cal crear un nou projecte de la mateixa forma que s'ha fet en les sessions anteriors. El circuit que cal entrar en l'esquemàtic és el de la figura 3.




Figura 3. Proposta d'esquemàtic per a la simulació de la característica I-V d'una cèl·lula solar.

Una cèl·lula solar és bàsicament una unió P-N (un díode ), d'on podem extreure un corrent fotogenerat a partir de la llum incident en el dispositiu. El díode de la figura 3 és aquesta unió P-N, les resistències R1 i R3 permeten modelar les pèrdues en el dispositiu, i la font de corrent controlada G1 ens permet simular el corrent fotogenerat, que és proporcional a la irradiància a la que treballa la cèl·lula solar. Com el corrent fotogenerat és proporcional a la irradiància, per això li assignem un valor de guany a la font controlada, mentre que per simular el valor de la irradiància s'utilitza una font de tensió V1, en aquest cas de valor 24 V. Finalment la font de tensió Vbias ens permetrà fer un escombrat de tensió i obtenir la característica I-V de la cèl·lula solar.

Un cop introduït l'esquemàtic, heu de configurar el perfil de la simulació amb: **Pspice > New Simulation Profile**, o bé clickant la icona equivalent del menú **Pspice**.




Figura 4. Finestra per definir el tipus d'anàlisi.

Després de donar nom al perfil, heu de configurar-lo amb una anàlisi del tipus **“DC Sweep”**, la variable d'escombrat **“Sweep variable”** a **“Voltage source”** i el paràmetre **“Name”** amb la font **Vbias**. L'inici de l'escombrat heu de posar-lo a **“Start Value”** de **-0.5 V**, el final a **“End Value”** igual a **1 V** i el pas de tensió a **“Increment”** igual a **0.01 V**. Tal com podeu veure en la figura 4.

Executem la simulació i anem a analitzar els resultats.

Fem **Trace > Add Trace**, o bé clicquem la icona equivalent del menú **Pspice** de la part superior de la pantalla. Aquesta acció fa obrir la finestra de diàleg que permet triar la variable que es vol visualitzar.

La variable que visualitzarem és el corrent  $I(V_{bias})$ . Per aquest motiu seleccionarem de la llista "**Simulation Output Variables**" la variable " **$I(V_{bias})$** " i fem "**OK**". Obtindrem un plot com el de la figura 5. On apareix la característica corrent-tensió de la cèl·lula solar.




Figura 5. Resultat de la simulació.

Fent un doble-clic sobre una de les escales de la gràfica que ens mostra el resultat de la simulació s'obre la finestra de definició dels eixos "**Axis Settings**" i podem fer un escalat amb la opció **Data Range > User defined** i analitzar millor els resultats. Feu un escalat de l'eix de corrent entre -25 mA i 25 mA i en l'eix de tensions entre -50 mV i 0,9 V.




Figura 6. Finestra Axis Settings.

Si fem un clic amb el cursor sobre els punts de la característica podrem utilitzar el cursor amb la opció "**Cursor on**".

Amb aquesta opció activada podeu desplaçar-vos per sobre d'aquesta característica i analitzar-ne els valors. Per exemple, si ens fixem a  $V = 0$  podem extreure el valor del corrent de curtcircuit,  $I_{sc}$ , de la cèl·lula solar, que en aquest cas és de 7,2 mA. Si busquem el punt on el corrent és zero podem obtenir el valor de la tensió de circuit obert,  $V_{oc}$ , que en aquest cas és de 715 mV.

Podem també fer un plot de la potència que pot donar el dispositiu en funció del seu punt de treball. Per analitzar-ho fem:

**Trace > Add Trace**

i triem a **Trace Expression:  $I(V_{bias}) * V_{V_{bias}}$**

Per buscar el punt de màxima potència en la característica obtinguda, fem un altre cop **“Cursor on”** i ara fem un clic en l'opció **“Cursor Peak”** de la barra de menú. Podem veure el valor d'aquest punt de màxima potència a la taula inferior o bé clicar al **“Mark Label”** de la barra de menú. El resultat és que el punt de màxima potència té coordenades (0.6 V, 3.9 mW) com podeu veure a la figura 7.




Figura 7. Punt de màxima potència.

Ara tornem a la finestra de OrCAD Capture. Podem escollir **Pspice > Create Netlist**, i amb aquesta opció el programa genera el fitxer de text corresponent a l'esquemàtic. Per veure el contingut d'aquest fitxer, fem ara **Pspice > View Netlist**. El programa “Bloc de Notes” ens mostrarà el text corresponent a aquest Netlist, que ha de ser semblant al següent:

```
* source {nom_fitxer}

R_R1      N00992 N00689 0.00005 TC=0,0

R_R3      0 N00992 1000k TC=0,0

G_G1      0 N00992 N00629 0 0.0003

V_V1      N00629 0 24V

V_Vbias   N00689 0 0Vdc

X_D1      N00992 0 D1N4500
```

Compareu el contingut del Netlist amb l'esquemàtic dibuixat a OrCAD Capture.

## Com podem fer una simulació a partir d'un fitxer de text o Netlist?

La simulació Pspice d'un circuit es pot fer a partir d'un esquemàtic dibuixat en OrCAD Capture o mitjançant la crida d'un fitxer Netlist directament des de l'entorn PspiceAD.

En aquest apartat triarem aquesta segona opció per realitzar una simulació molt semblant a la realitzada en el punt anterior per a una cèl·lula solar.

El primer pas és la creació del Netlist, que ha de ser un fitxer de text amb tota la informació corresponent al circuit que volem simular. Per fer-ho, obrim el programa "Bloc de Notas", on cal escriure el següent text:

```
* model basic cel.lula solar

* Constant de Boltzman
.param k=1.38e-23
* carrega electro
.param q=1.6e-19

* definicio subcircuit i valors per defecte dels principals parametres
.subckt cell 5 4 3 2 PARAMS: Area=1, Jsc=1, Voc=1, Rs=1, Rp=1, nd=1

girrad 4 1 value={{(Jsc/1000)*v(3)*Area}
rs 1 2 {Rs}
rp 1 4 {Rp}
d1 1 4 diode
.model diode d( is={{(Jsc*Area)/(exp(Voc*q/(nd*300*k))-1)}, n={nd}})
virrad 3 4 dc 1000
.ends cell

* Primera simulacio
.temp=27

xcell 1 0 2 3 cell params: Area=1 Rs=167 Rp=11720.46 nd=1.48 Jsc=0.006 Voc=0.64
vpolaritzacio 3 0 dc 0
rpot 2 0 100000
.plot dc i(vpolaritzacio)
.probe
.dc vpolaritzacio -1.5 1 0.01
.end
```

Salveu el circuit amb el nom **1Sim.cir**.

## Comentaris sobre el netlist

---

- Totes les sentències que van precedides d'un asterisc (\*) són comentaris i no són interpretats pel programa PspiceAD.
- La sentència **.param**, ens permet definir constants. En aquest cas n'hi ha dues: La constant de Boltzman i la càrrega de l'electró. Un cop definides poden ser utilitzades a la resta del Netlist amb el valor que tenen assignat.
- En el Netlist es defineix un subcircuit, sentència **.subck**. Això també es pot fer amb el programa OrCAD Capture. A la definició del subcircuit apareix el nom "cell", seguit dels nodes d'accés al subcircuit (entrades i/o sortides), que en aquest cas són 4 nodes, i finalment els paràmetres associats al subcircuit. Els paràmetres definits en aquest subcircuit són els següents:
  - Area : àrea de la cèl·lula solar ( $\text{cm}^2$ ).
  - Jsc : densitat de corrent de curtcircuit de la cèl·lula solar ( $\text{A/cm}^2$ ).
  - Voc : tensió de circuit obert de la cèl·lula solar (V).
  - Rs i Rp : resistències de pèrdues associades al dispositiu ( $\Omega$ ).

A cada paràmetre se li assigna un valor per defecte, en aquest cas tots valen 1.

- En el subcircuit cell tenim els següents elements :
  - Font de corrent controlada per tensió: *girrad*, connectada entre els nodes 1 i 4. El valor del corrent es defineix mitjançant la sentència **Value = {...}**, que en aquest cas és funció del valor de la tensió d'entrada al node 3 (V(3)), la densitat de corrent de curtcircuit (Jsc), i l'àrea del dispositiu.
  - Resistència **rs**, entre els nodes 1 i 2, de valor Rs i resistència **rp** de valor RP entre els nodes 1 i 4.
  - El díode **d1** entre els nodes 1 i 4. Aquest díode es defineix mitjançant la crida al seu model ( sentència **.model** ) On podem modificar algun dels paràmetres d'aquest model. En aquest cas el factor de idealitat del díode : **n** prendrà el valor que assignem al paràmetre **nd**, i el corrent invers de saturació del díode, **is**, ve donat per l'expressió que apareix en la sentència. Això permet redefinir paràmetres de models existents per fer simulacions més acurades.
  - Una font de tensió continua DC: **virrad** , entre els nodes 3 i 4 de valor 1000, que es correspon amb una irradiància de 1000 ( $\text{W/m}^2$ ).
- Per finalitzar el subcircuit cal escriure la sentència : **.ends** ( nom del subcircuit), en aquest cas **.ends cell**.
- Hem definit el subcircuit dintre del fitxer que utilitzarem per fer la simulació per comoditat. Aquesta definició de subcircuit podria fer-se en un fitxer de text independent amb extensió **.lib**. En aquest cas només cal fer la crida al subcircuit en el nostre Netlist de simulació. Aquesta crida es defineix tot seguit.
- Abans de fer la crida al subcircuit definim un temperatura de treball amb la sentència **.temp = {...}**, en aquest cas 27°C. Això ho fem perquè algun model dels utilitzats, en aquest cas el díode, pot necessitar aquest paràmetre per fer algun càlcul intern.
- La part principal del netlist comença amb la crida al subcircuit, X seguit de nom de subcircuit, en aquest cas **xcell**, seguidament es reassignen els nodes del



subcircuit a la nova numeració. Aquesta numeració de nodes assigna el nou número de node al node del subcircuit que ocupa el mateix lloc en la sentència de definició del subcircuit.

- Seguidament donem valor als paràmetres interns del subcircuit. Aquest valor són els que s'utilitzaran en la simulació.
  - Per veure la característica corrent-tensió de la cèl·lula solar afegim ara una font de tensió DC amb la seva resistència sèrie, **Vpolarització** i **rpol**.
  - Indiquem que cal executar el probe amb la sentència **.probe**, això farà obrir el menú gràfic del pspiceAD un cop finalitzada la simulació. També indiquem que directament ens faci un plot de la variable **I(Vpolarització)** amb la sentència **.plot I(Vpolarització)**.
  - Per indicar el tipus d'anàlisi que ha de fer el programa afegim la sentència: **.dc vpolaritzacio -1.5 1 0.01**, això indica que ha de fer un escombrat DC de la font Vpolarització de -1.5 V a 1 V amb un pas de 0.01 V.
  - Finalment cal tancar el fitxer amb la sentència **.end**.
- 

Ara farem la simulació d'aquest netlist amb el programa PspiceAD.

Obriu el programa Pspice AD:

**Inicio > Todos los programas > Cadence > Release 16.3 > Pspice AD**

Feu **File > Open > 1Sim.cir**. Ens apareixerà a la finestra de PSpiceAD el fitxer que hem creat anteriorment. A la barra de menú trieu **Simulation > Run 1Sim**.

Es realitzarà la simulació i s'obrirà l'entorn gràfic d'anàlisi de resultats. Feu **Trace > Add trace > I(Vpolarització)**. Obtindreu el resultat que mostra la figura 8.

Mitjançant els cursors indiqueu el punt de màxima potencia, el valor de la corrent de curtcircuit i el valor de la tensió de circuit obert.




Figura 8

---

## Exercici 1

Dibuixeu en un paper el circuit corresponent a l'esquemàtic del netlist amb el que hem fet la simulació Sim1.cir. Indiqueu la numeració dels nodes i compareu aquest esquemàtic amb el del circuit de la figura 3.

¿Quines són les principals diferències entre tots dos circuits ?

---

---

## Exercici 2

El corrent de curtcircuit d'una cèl·lula solar és pràcticament proporcional a la irradiància. Les variacions en aquest paràmetre han de traduir-se en variacions equivalents a  $I_{sc}$ , mentre que la variació de  $V_{oc}$  és logarítmica.

Heu de fer les simulacions amb el model de la cèl·lula solar presentat abans amb diferents valors de la irradiància :  $1500 \text{ W/m}^2$ ,  $700 \text{ W/m}^2$  i  $250 \text{ W/m}^2$ . Podeu fer això amb una única simulació. Cal definir un paràmetre (IRRAD) per simular la irradiància. Podeu fer aquestes modificacions en el fitxer anterior d'entrada a PspiceAD:

\* subcircuit cell

params : IRRAD=1  
Virrad 3 4 dc IRRAD

\* simulacio de tot el circuit

.param IR=1  
xcell 1 0 2 3 cell params: IRRAD=IR (\*i la resta de paràmetres\*)  
.step param IR list 1500 700 250

Per a cada una de les simulacions realitzades amb diferent valor del paràmetre irradiància, doneu els valors obtinguts de :

Irradiància ( $\text{W/m}^2$ )	1500	700	250
Isc (mA)			
Voc (V)			

Realitzeu les simulacions amb el model de la cèl·lula solar que s'ha presentat abans per a diferents valors de  $R_s$ :  $1 \text{ m}\Omega$ ,  $0.3 \Omega$  i  $2 \Omega$ . Agafeu un valor constant per a la irradiància ( $1000 \text{ W/m}^2$ ). Per a cadascuna de les simulacions doneu els valors obtinguts d'aquests paràmetres :

$R_s$	$0.3 \Omega$	$1 \Omega$	$2 \Omega$
Isc (mA)			
Voc (V)			
Punt de màxima potència (W)			

---

## Com es pot crear un símbol per definir un subcircuit a OrCAD Capture?

La simulació que hem fet en l'anterior apartat a partir d'un fitxer Netlist d'entrada a PSpice AD la podem repetir mitjançant la creació d'un símbol pel subcircuit i treballant amb l'entrada d'esquemàtics de OrCAD Capture.

Per fer-ho cal seguir les següents indicacions:

- Amb el bloc de Notes s'ha de crear el fitxer i guardar-lo a la llibreria de PSpice amb el nom **Scell.mod**:

```
* model basic cel.lula solar
* definicio subcircuit i valors per defecte dels principals parametres
.subckt cell 1 2 3 PARAMS: Area=1, Jsc=1, Voc=1, Rs=1, Rp=1, nd=1
* Constant de Boltzman
.param k=1.38e-23
* carrega electro
.param q=1.6e-19
girrad 3 5 value={{(Jsc/1000)*v(3)*Area}
rs 5 2 {Rs}
rp 5 3 {Rp}
d1 5 3 diode
.model diode d( is={{(Jsc*Area)/(exp(Voc*q/(nd*300*k))-1)}, n={nd})
.ends cell
```

- Obriu el PSpice Model Editor amb:  
**Inicio > Todos los programas > Cadence > Release 16.3 > Pspice Accessories > Model Editor**

A la finestra d'entrada "**Select Design Entry tool**", trieu **Capture**.

Seguidament feu : **FILE > NEW**. Ara aneu a **MODEL > IMPORT** i busqueu el model pel que voleu crear el Símbol, en aquest cas el fitxer **Scell.mod**. Això obrirà el fitxer. Salveu aquest fitxer amb extensió **.lib** al directori on teniu la resta dels fitxers de la vostra simulació

El pas següent es crear el component "Part" corresponent per utilitzar a Capture. Amb el fitxer **Scell.lib** encara obert anar a **FILE > Export to Capture Part library**, s'obriran unes finestres que us permeten crear el fitxer i salvar-lo en el directori corresponent. Això farà que tingueu accessible el fitxer amb extensió **.olb**.

Ara ja podem editar el Part Symbol que hem creat i crear un nou projecte a OrCAD Capture on podrem entrar-lo a l'esquemàtic i treballar amb el nou símbol.

Obriu el programa OrCad Capture per crear un nou projecte de la mateixa forma que s'ha fet en les sessions anteriors. Doneu nom al vostre projecte.

Aneu al menú: **FILE>OPEN>LIBRARY** i busqueu el fitxer al directori on l'hem salvat anteriorment. Ara podem editar el component i el símbol que hem creat a OrCAD Capture. Si feu un click a **CELL > Edit Part**, podreu veure el nou símbol i els nodes d'entrada i/o sortida.

El circuit que cal entrar en l'esquemàtic és el de la figura 9. Recordeu que cal afegir la llibreria **Scell.lib** per poder triar el component **cell**.




Figura 9. Esquemàtic per a la simulació.

Si feu doble clic al símbol de la pàgina de l'esquemàtic podreu introduir els valors dels paràmetres corresponents. Introduïu els següents valors dels paràmetres a la finestra que mostra la figura 10:

*PARAMS: Area=1, Jsc=0.0061, Voc=0.6, Rs=1, Rp=10000, nd=1*




Figura 10. Finestra per introduir els valors dels paràmetres del símbol.

Un cop introduït l'esquemàtic, heu de configurar el perfil de la simulació amb: **Pspice > New Simulation Profile**, o bé clickant la icona equivalent del menú **Pspice**. Després de donar nom al perfil, heu de configurar-lo amb una anàlisi del tipus “**DC Sweep**”, la variable d'escombrat “Sweep variable” és una “**Voltage Source**” i el paràmetre “Name” **V1**. L'inici de l'escombrat heu de posar-lo a “Start Value” de **-1 V**, el final a “End Value” igual a **1 V** i el pas de tensió a “Increment” igual a **0.05 V**.

Ara cal que afegim la nova llibreria a **Configuration Files**, una pestanya de la finestra Simulation Settings on estem definint el perfil de la simulació. Un cop allà, s'ha de triar a “**Category: Library**”, amb el Browse s'ha de buscar la llibreria corresponent a “**Scell.lib**” i afegir-la al disseny amb “**Add to design**”.

Ara ja podem engegar la simulació Pspice. Feu un plot de I(V1) i compareu els valors de la tensió de circuit obert, corrent de curtcircuit i punt de màxima potència de la cèl·lula solar. Compareu aquests valors amb els dels paràmetres que s'han definit pel nostre símbol/part de la cèl·lula solar.

---

### **Exercici 3**

Realitzeu ara l'exercici 2 a partir del programa OrCAD Capture mitjançant l'esquemàtic d'entrada. Indiqueu els resultats obtinguts.

---

## **Modulació d'amplitud, AM Modulated Signal**

Un senyal modulat en amplitud, (AM), pot definir-se amb la següent expressió :

$$v_{am}(t) = [(A + V_m \cos(2\pi f_m t))] \cos(2\pi f_c t) = A[1 + m \cos(2\pi f_m t)] \cos(2\pi f_c t) \quad (1)$$

en la que una portadora sinusoidal d'alta freqüència:  $\cos(2\pi f_c t)$  es modulada per una moduladora sinusoidal de freqüència  $f_m$ . El senyal modulador pot ser de qualsevol tipus, en aquest exemple agafarem un senyal sinusoidal, i a l'índex de modulació l'anomenem  $m$ .




Figura 11. Esquemàtic per generar un senyal modulat AM.

Per implementar un senyal modulat en amplitud, AM, en PSpice podem utilitzar la funció MULTIPLICAR que tenim disponible a la llibreria "function.olb", que teniu en el directori **tools > capture > library > pspice > advanls**. Caldrà afegir aquesta llibreria al menú d'esquemàtic de OrCAD Capture i triar la part **Multiplier 2** que realitza el producte de les seves dues entrades. La figura 11 mostra l'esquemàtic que genera el senyal modulat AM sobre una resistència R1.

Ara cal realitzar una simulació Pspice per veure el senyal modulat. Amb aquest objectiu cal que definiu a **Simulation Settings** un anàlisi tipus **Time domain Transient**, amb un temps de simulació de 10 ms. El resultat obtingut serà el que mostra la figura 11.




Figura 12. Forma d'ona simulada (transient analysis) del circuit de la figura 9, amb (A=1V,  $f_m=500$  Hz,  $f_p=5$ kHz i  $m=0.5$ )

Podem també veure el contingut espectral d'aquest senyal mitjançant la transformada de Fourier que ens ofereix la opció **Trace > Fourier** del menú de sortida de simulacions Pspice o bé amb la icona **FFT**.

Podeu canviar el marge de l'eix X de freqüència fent doble-click a sobre de l'eix. La figura 13 mostra l'espectre del senyal modulad, amb un pic màxim corresponent al senyal portador de 5kHz i dos pics laterals a 4.5 i 5.5 kHz, que indiquen que la freqüència del senyal modulador és de 500Hz. Podeu utilitzar els cursors per tenir millor resolució de lectura.




Figura 13. Contingut espectral del senyal de la figura 11.

---

## Exercici 4

Realitzeu una simulació Pspice per veure el senyal modulad AM. Utilitzeu una portadora sinusoidal de freqüència 120kHz i un senyal modulador sinusoidal de freqüència  $f_m = 1$ kHz.

Visualitzeu el senyal modulad i obtingueu informació del seu contingut espectral mitjançant l'anàlisi de Fourier de Pspice.

---